

Problem 1a:

Let's first define the two differential equations and initial conditions that define the Lotka Volterra model:

$equations := \{diff(m(t), t) = (a - \text{gamma} \cdot w(t)) \cdot m(t), m(0) = m_0, diff(w(t), t) = (\text{delta} \cdot m(t) - \text{epsilon}) \cdot w(t), w(0) = w_0\};$

$$\left\{ m(0) = m_0, w(0) = w_0, \frac{d}{dt} m(t) = (a - \gamma w(t)) m(t), \frac{d}{dt} w(t) = (\delta m(t) - \epsilon) w(t) \right\} \quad (1)$$

Then define a set of values for the various coefficients in the model (this is the first such set, we will modify them later on for the other parts of the problem):

$coefficients1 := \{a = 0.4, \text{gamma} = 0.002, \text{delta} = 0.0025, \text{epsilon} = 0.1, m_0 = 1000, w_0 = 50\};$

$$\{a = 0.4, \delta = 0.0025, \epsilon = 0.1, \gamma = 0.002, m_0 = 1000, w_0 = 50\} \quad (2)$$

Let us now also define a set of equations where we have replaced the coefficients by their numerical values:

$ode1 := subs(coefficients1, equations);$

$$\left\{ m(0) = 1000, w(0) = 50, \frac{d}{dt} m(t) = (0.4 - 0.002 w(t)) m(t), \frac{d}{dt} w(t) = (0.0025 m(t) - 0.1) w(t) \right\} \quad (3)$$

The next step is to numerically solve this set of equations:

$solution1 := dsolve(ode1, numeric, output = listprocedure);$


$$[t = \mathbf{proc}(t) \dots \mathbf{end proc}, m(t) = \mathbf{proc}(t) \dots \mathbf{end proc}, w(t) = \mathbf{proc}(t) \dots \mathbf{end proc}] \quad (4)$$

The output of all this is a vector of three objects; the second and third are the two functions $m(t)$ and $w(t)$ that we are interested in. For example, if we look at the second element of this vector we get $solution1[2];$


$$m(t) = \mathbf{proc}(t) \dots \mathbf{end proc} \quad (5)$$

Note that this is an equation that relates the left and the right hand side. It is not a function of t . But we can get a function of t the following way, taking the right hand side of the equality and evaluating it at time t :

$plot(rhs(solution1[2])(t), t = 0 .. 1000);$




This also allows us to plot both moose and wolves at the same time:
`plot({rhs(solutionI[2])(t), rhs(solutionI[3])(t) }, t=0..1000);`



We observe a boom-and-bust behavior: everytime the number of wolves (green curves) becomes small, the moose population grows exponentially but then the wolves catch up and quickly decimate the available moose supply until they eventually succumb to hunger again, repeating the whole cycle. It is quite obvious that we haven't chosen our parameters in any useful way at all: the wolf population grows to twice the maximal moose population. This is certainly not reasonable and is due to an improper set of coefficients. A more sophisticated analysis of the system would reveal which variables we have to tweak to get a more realistic set of coefficients.


The problem also asked for a phase plot of this system. This can be achieved as follows (read up on the `odeplot` function in Maple's help pages):

```
plots[odeplot](solution1, [m(t), w(t)], t=0..1000);
```




This doesn't quite look right: One would imagine that the trajectory of the system traces out a smooth curve, rather than the spikes we can see here as the growth rate (i.e. the derivatives) of moose and wolf populations jump abruptly. The spikes are in fact only an artifact of how Maple plots the trajectories: it computes a data point every few years and then draws a straight line between each two. To get a smoother curve, we need to tell Maple to reduce the length of the interval between each two points a bit (or alternatively: plot more points). We can get a better plot like this, asking Maple to plot one point per year (1000 points for the 1000 year time interval):

```
plots[odeplot](solution1, [m(t), w(t)], t=0..1000, numpoints = 1000);
```



We get an even better plot by asking for one point every day:
`plots[odeplot](solution1, [m(t), w(t)], t=0..1000, numpoints = 365000);`



Problem 1b:

In the second part, we need to play a bit with the lifetime of moose. In the first part, we used a lifetime of 5 years, meaning that the difference a between birth rate and death rate was $a = \frac{1}{2} - \frac{1}{5} = 0.4$. With a

lifespan of 2.5 years, the coefficient would now take the value $a = \frac{1}{2} - \frac{1}{2.5} = 0.1$. We can solve the respective equations as follows now:


$coefficients2 := \{a = 0.1, \text{gamma} = 0.002, \text{delta} = 0.0025, \text{epsilon} = 0.1, m_0 = 1000, w_0 = 50\} :$

$ode2 := \text{subs}(coefficients2, equations);$

$$\left\{ m(0) = 1000, w(0) = 50, \frac{d}{dt} m(t) = (0.1 - 0.002 w(t)) m(t), \frac{d}{dt} w(t) = (0.0025 m(t) - 0.1) w(t) \right\} \quad (6)$$

$solution2 := \text{dsolve}(ode2, \text{numeric}, \text{output} = \text{listprocedure}) :$

$\text{plot}(\{rhs(solution2[2])(t), rhs(solution2[3])(t)\}, t = 0 .. 1000);$



The phase plot does look similar to the one we had before, but we can see that while qualitatively the same, with the new coefficient values we now get fewer boom-and-bust cycles within the 1000 year time step. A similar change can be seen when making the lifespan of moose much larger, to 100 years. This

leads to $a = \frac{1}{2} - \frac{1}{100} = 0.49$ and the following graphs:


`coefficients3 := {a = 0.49, gamma = 0.002, delta = 0.0025, epsilon = 0.1, m0 = 1000, w0 = 50} :`

`ode3 := subs(coefficients3, equations);`

$$\left\{ m(0) = 1000, w(0) = 50, \frac{d}{dt} m(t) = (0.49 - 0.002 w(t)) m(t), \frac{d}{dt} w(t) = (0.0025 m(t) - 0.1) w(t) \right\} \quad (7)$$

`solution3 := dsolve(ode3, numeric, output = listprocedure) :`

`plot({rhs(solution3[2])(t), rhs(solution3[3])(t)}, t = 0 .. 1000);`



This time the time period between two boom-and-bust cycles has grown shorter, compared to the original model.

Problem 1c:

The third part of the problem deals with the question whether the answers make quantitative sense. One may agree that the answers make qualitative sense: boom-and-bust cycles are frequently observed in nature, but do the numbers make sense? To investigate this, let's take a look at the minimal number of moose for each of the three cases we considered above:

Optimization[Minimize](rhs(solution1[2])(t), t=0..1000);

$$[8.32902911918957 \cdot 10^{-10}, [t=960.214534704332]] \quad (8)$$

Optimization[Minimize](rhs(solution2[2])(t), t=0..1000);

$$[1.37296953361324 \cdot 10^{-8}, [t=330.779090642298]] \quad (9)$$

Optimization[Minimize](rhs(solution3[2])(t), t=0..1000);

$$[2.03787843597333 \cdot 10^{-10}, [t=394.935304864204]] \quad (10)$$

So what this shows is that at various times in the cycles, the moose number sometimes drops to a very small fraction of a single moose, from which the population then miraculously recovers. This is clearly not physical: to recover, a population needs to have not just one but at the very least two members, and we are far below that threshold. What we can conclude from this is that the model is definitely not a

particularly good approximation of reality if populations become very small. That may not surprise: in the derivation of the model we have not made any accommodations to the fact that animal populations need to be numerous enough so that individuals can, for example, meet and reproduce. The right approach would therefore be to note the deficiencies of this model and, if small populations are important to the person who is interested in the model, make modifications to the model that take into account the fact that populations are whole numbers, for example.

More discussions:

One thing we may want to consider is what happens if we do not deviate that far from the equilibrium points. To this end, let us look at the equations and note that if the term in parentheses are zero, then population numbers stay constant:

equations;

$$\left\{ m(0) = m_0, w(0) = w_0, \frac{d}{dt} m(t) = (a - \gamma w(t)) m(t), \frac{d}{dt} w(t) = (\delta m(t) - \epsilon) w(t) \right\} \quad (11)$$

In other words, if we start at $w_0 = \frac{a}{\gamma}$ and $m_0 = \frac{\epsilon}{\delta}$ then the rates of change are zero and populations stay constant. So let's solve our differential equations with these values as starting points:


$$\text{coefficients4} := \left\{ a = 0.4, \gamma = 0.002, \delta = 0.0025, \epsilon = 0.1, m_0 = \frac{0.1}{0.0025}, w_0 = \frac{0.4}{0.002} \right\} :$$

ode4 := subs(coefficients4, equations);

$$\left\{ m(0) = 40.00000000, w(0) = 200.0000000, \frac{d}{dt} m(t) = (0.4 - 0.002 w(t)) m(t), \frac{d}{dt} w(t) = (0.0025 m(t) - 0.1) w(t) \right\} \quad (12)$$

solution4 := dsolve(ode4, numeric, output = listprocedure) :

plot({rhs(solution4[2])(t), rhs(solution4[3])(t)}, t = 0 .. 1000);



OK, so the population stays constant now at 40 moose and 200 wolves (the fact that these numbers by themselves don't make sense is a consequence of the particular set of coefficients we have chosen, which apparently don't make much physical sense). Now, what if we slightly perturb this equilibrium by adding one to the existing 40 moose:


$$\text{coefficients5} := \left\{ a = 0.4, \text{gamma} = 0.002, \text{delta} = 0.0025, \text{epsilon} = 0.1, m_0 = \frac{0.1}{0.0025} + 5, w_0 = \frac{0.4}{0.002} \right\};$$

`ode5 := subs(coefficients5, equations);`


$$\left\{ m(0) = 45.00000000, w(0) = 200.0000000, \frac{d}{dt} m(t) = (0.4 - 0.002 w(t)) m(t), \frac{d}{dt} w(t) = (0.0025 m(t) - 0.1) w(t) \right\} \quad (13)$$

`solution5 := dsolve(ode5, numeric, output = listprocedure) :`

`plot({rhs(solution5[2])(t), rhs(solution5[3])(t)}, t = 0 .. 1000);`



```
plots[odeplot](solution5, [m(t), w(t)], t=0..1000, numpoints = 365000);
```



This is more like what we expected: A small perturbation of the system leads to a small oscillation, clearly visible in both the time dependent plot as well as the phase plot (note the scales of the phase plot) . Feel free to play around a bit and see what happens if you make the initial perturbation (here: 5 moose) larger and larger!


Since we have been analyzing the trajectory of our system in phase plot, I wanted to point you at one other, nifty tool for showing what is going on: Given our system of two equations without initial conditions

equationonly := [diff(m(t), t) = (a - gamma*w(t)) * m(t), diff(w(t), t) = (delta*m(t) - epsilon) * w(t)];

$$\left[\frac{d}{dt} m(t) = (a - \gamma w(t)) m(t), \frac{d}{dt} w(t) = (\delta m(t) - \epsilon) w(t) \right] \quad (14)$$


and given the solution we had for the initial problem, we can show the phase plot also with a different command:

DEtools[phaseportrait](subs(coefficients1, equationonly), [m(t), w(t)], t=0..1000, [[m(0) = 1000, w(0) = 50]], numpoints = 10000, linecolor = blue);



This shows the same curve as the original phase plot but it now also has little arrows that show which direction the system would move in if we started at a given point. The trajectory clearly follows curves described by these arrows. We can of course do the same again looking at only the area around the little perturbation we looked at just above:

```
DEtools[phaseportrait](subs(coefficients1, equationonly), [m(t), w(t)], t=0..1000, [[m(0)
= 0.1/0.0025 + 5, w(0) = 0.4/0.002 ]], numpoints = 10000, linecolor = blue);
```



This is what we've already seen in class: in the vicinity of the stationary point, perturbations run around in circles.